

# CS421 Fall 2011 Midterm 1

Tuesday, October 12, 2011

Name:	
NetID:	

- You have **75 minutes** to complete this exam.
- This is a **closed-book** exam. You are allowed one 3inch by 5inch card of notes prepared by yourself. This card is **not to be shared**. All other materials, besides pens, pencils and erasers, are to be away.
- Do not share anything with other students. Do not talk to other students. Do not look at another student's exam. Do not expose your exam to easy viewing by other students. Violation of any of these rules will count as cheating.
- If you believe there is an error, or an ambiguous question, you may seek clarification from myself or one of the TAs. You must use a whisper, or write your question out. Speaking out aloud is not allowed.
- Including this cover sheet and rules at the end, there are 9 sheets, 17 pages to the exam, including one blank page for workspace. The exam is printed double sided. Please verify that you have all 17 pages.
- Please write your name and NetID in the spaces above, and also at the top of every sheet.

<b>Problems</b>	<b>Possible Points</b>	<b>Points Earned</b>
1	6	
2	8	
3	11	
4	11	
5	11	
6	10	
7	11	
8	12	
9	20	
<b>PreTotal</b>	<b>100</b>	
<b>Extra Credit</b>	<b>10</b>	
<b>PostTotal</b>	<b>110</b>	

CS 421 Midterm 1

Name: \_\_\_\_\_

1. (6 pts total) Suppose that the following code is input one line at a time into OCaml:

```
let z = 32.0;;  
let r = 1.8;;  
let trans x z = (x *. r) +. z;;  
let z = 4.0;;  
let r = 5.0;;  
let a = trans 2.0 1.0;;  
let b = trans 1.0;;  
let c = trans (1.0, 1.0);;
```

For each of **a**, **b**, and **c**, either give what result is returned, or give the reason why nothing is returned.

a. (2 pt) Tell what is returned, if anything, for **a**, or why not:

**Solution: 4.6**

b. (2 pt) Tell what is returned, if anything, for **b**, or why not:

**Solution: a function from floats to floats, adding 1.8 to its input**

c. (2 pt) Tell what is returned, if anything, for **c**, or why not:

**Solution: Attempting to evaluate c causes a type error, because trans take as a argument an int, but it is being applied to an (int \* int)**

2. (8 pts total) For each of the following programs, indicate what is printed, and value is returned in each case:

a. (4pts) `let x = (print_string "a"; 5)  
in (fun y -> (print_string "b"; (y *x) + y)) (print_string "c"; 3);;`

**Solution:** prints: acb      returns: 18

b. (4 pts) `(fun z -> (print_string "a"; ((z() * 5) + z())) (fun () -> (print_string "b"; 3)));;`

**Solution:** prints: abb      returns: 18

CS 421 Midterm 1

Name: \_\_\_\_\_

3. (11 pts total) Consider the following OCaml code

```
let x = 5;;  
let a = x + 4;;  
let f y z = let h x = x + a in y + h z;;  
let a = 20;;  
let x = f a 3;;
```

Describe the final environment that results from the execution of the above code if execution is begun in an empty environment. Your answer should be written as a set of bindings of variables to values, with only those bindings visible at the end of the execution present. Your answer should be a precise mathematical answer, with a precise description of values involved in the environment. The update operator (+) and abbreviations should not be used.

**Solution:** $\{f \rightarrow \langle y, \text{fun } z \rightarrow \text{let } h \text{ } x = x + a \text{ in } y + h \text{ } z, \{x \rightarrow 5; a \rightarrow 9\} \rangle, a \rightarrow 20, x \rightarrow 32\}$

4. (11 pts) Write a function **partial\_sums : float list -> float list** that takes a list of floats  $[x_0; \dots; x_n]$  and returns a list whose  $i^{\text{th}}$  element is the sum of the elements in positions  $i$  to  $n$ ,  $\sum_{j=i}^n x_j$ , for each  $i = 0, \dots, n$ . A sample execution is as follows:

```
# let rec partial_sums lst = ...
val partial_sums : float list -> float list = <fun>
# partial_sums [4.0; 2.2; 7.5];;
- : float list = [13.7; 9.7; 7.5]
```

You are allowed to start your code with **let rec**, but are not required to. You are allowed to use any auxiliary functions you write yourself here, but you are **not** allowed to use any library functions. You may use the append function **@**.

### Solution:

```
let rec partial_sums lst =
  match lst with [] -> []
  | x::xs ->
    (match partial_sums xs with [] -> [x]
     | y::ys -> (x +. y) :: (y :: ys))
```

CS 421 Midterm 1

Name: \_\_\_\_\_

5. (11 pts total)

- a. (5 pts) Write a function **count\_if** : ('a -> bool) -> 'a list -> int such that **count\_if p lst** returns the number of elements in **lst** for which the given predicate **p** gives **true**. The function is required to use only tail recursion (no other form of recursion). You may use auxiliary functions, but they must also use no other form of recursion than tail recursion, You may **not** use any library functions. Executing your code should give the following behavior:

```
# let rec count_if p lst = ... ;;
val count_if : ('a -> bool) -> 'a list -> int = <fun>
# count_if (fun x -> x > 3) [1;2;3;4;5];;
- : int = 2
```

**Solution:**

```
let rec count_if p lst =
  let rec count_if_aux p lst count =
    (match lst with [] -> count
     | (x::xs) -> count_if_aux p xs (if p x then 1+ count else count))
  in count_if_aux p lst 0
```

- b. (6 pts) Rewrite **count\_if** as described above, using

**List.fold\_left** : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a

but **no other** library functions and **no explicit recursion**.

**Solution:**

```
let count_if p lst =
  List.fold_left (fun count -> fun x -> if p x then 1 + count else count) 0 lst
```

6. (10 pts) Consider the following code:

```
let rec map f l =
  match l with [] -> [] | (x :: xs) -> (f x) :: map f xs ;;
```

- a. (2 pts) Write **consk**, the Continuation Passing Style version of **::**. Your function should have the following type:

```
consk : 'a -> 'a list -> ('a list -> 'b) -> 'b
```

**Solution:** let consk x xs k = k(x :: xs)

- b. (8 pts) Write a function **mapk** that is a complete Continuation Passing Style transformation of **map**. Your function should have the following type:

```
mapk : ('a -> ('b -> 'c) -> 'c) -> 'a list -> ('b list -> 'c) -> 'c
```

**Solution:**

```
let rec mapk fk l k =
  match l with [] -> k []
  | x :: xs -> fk x (fun r1 -> mapk fk xs (fun r2 -> consk r1 r2 k))
```

Also acceptable:

```
let rec mapk fk l k =
  match l with [] -> k []
  | x :: xs -> mapk fk xs (fun r1 -> fk x (fun r2 -> consk r2 r1 k))
```



CS 421 Midterm 1

**Name:** \_\_\_\_\_

7. (11 pts). Create an OCaml recursive data type to describe propositional formulae made from true, false, propositional variables (named by strings), negation (not A), conjunctions (A and B), and disjunctions (A or B). Your data type should be able to model every propositional formulae described above, and nothing else.

**Solution:**

type prop = True | False | PropVar of string | Not of prop | And of prop \* prop | Or of prop \* prop

8. (12 pts total) Consider the following Ocaml recursive data types:

```

type const = Int of int | Bool of bool
type exp = VarExp of string
         | ConExp of const
         | IfExp of exp * exp * exp
         | AppExp of exp * exp
         | FunExp of string * exp

```

Write a function **num\_of\_consts : exp -> int** that counts the number of occurrences for the constructor **ConExp** in an **exp**. You may use recursion, auxiliary functions, and library functions from OCaml freely. A sample execution is as follows:

```

# let rec num_of_consts exp = ...
val num_of_consts : exp -> int = <fun>
# num_of_consts(IfExp (ConExp (Bool true), FunExp (“x”, ConExp (Int 5)), VarExp “y”));;
- : int = 2

```

**Solution:**

```

let rec num_of_consts exp =
  match exp
  with VarExp _ -> 0
       | ConExp c -> 1
       | IfExp (e1,e2,e3) -> (num_of_consts e1) + (num_of_consts e2) + (num_of_consts e3)
       | AppExp (e1,e2) -> (num_of_consts e1) + (num_of_consts e2)
       | FunExp (x, e) -> (num_of_consts e)

```

CS 421 Midterm 1

**Name:** \_\_\_\_\_

Worksheet (If extra space is needed).

9. (20 pts total) Give a type derivation for the following type judgment:

$\{\} \vdash \text{let } x = 5 > 3 \text{ in } ((\text{if } x \text{ then } (\text{fun } x \rightarrow x + 2) \text{ else } (\text{fun } y \rightarrow y)) 7) : \text{int}$

You may use the attached sheet of typing rules. Label every use of a rule with the rule used. You may abbreviate, but you must define your abbreviations. You may find it useful to break your derivation into pieces. If you do, give names to your pieces, which you may then use in describing the whole. Your environments should be mathematical mappings here, and NOT implementations as you might find in a program.

**Solution:**

$$\begin{array}{c}
 \text{Var} \text{-----} \quad \text{Con} \text{-----} \\
 \{x:\text{int}\} \vdash x:\text{int} \quad \{x:\text{int}\} \vdash 2:\text{int} \\
 \text{ArithOp} \text{-----} \quad \text{Var} \text{-----} \\
 \{x:\text{int}\} \vdash x + 2 : \text{int} \quad \{y:\text{int}, x:\text{bool}\} \vdash y:\text{int} \\
 \text{Fun} \text{-----} \quad \text{Fun} \text{-----} \\
 \text{Var} \text{-----} \quad \{x:\text{bool}\} \vdash \quad \{x:\text{bool}\} \vdash \\
 \{x:\text{bool}\} \vdash \text{fun } x \rightarrow x + 2 : \quad \text{fun } y \rightarrow y : \\
 x:\text{bool} \quad \text{int} \rightarrow \text{int} \quad \text{int} \rightarrow \text{int} \\
 \text{If} \text{-----} \\
 \text{Con} \text{-----} \quad \text{Con} \text{-----} \quad \{x:\text{bool}\} \vdash (\text{if } x \text{ then } (\text{fun } x \rightarrow x + 2) \quad \text{Con} \text{-----} \\
 \{x:\text{int}\} \vdash 5 > 3 : \text{bool} \quad \{x:\text{int}\} \vdash 7 : \text{int} \quad \text{else } (\text{fun } y \rightarrow y)) : \text{int} \rightarrow \text{int} \quad \{x:\text{bool}\} \vdash 7 : \text{int} \\
 \text{Rel} \text{-----} \quad \text{App} \text{-----} \\
 \{x:\text{int}\} \vdash 5 > 3 : \text{bool} \quad \{x:\text{bool}\} \vdash ((\text{if } x \text{ then } (\text{fun } x \rightarrow x + 2) \text{ else } (\text{fun } y \rightarrow y)) 7) : \text{int} \\
 \text{LetRule} \text{-----} \\
 \{\} \vdash \text{let } x = 5 > 3 \text{ in } ((\text{if } x \text{ then } (\text{fun } x \rightarrow x + 2) \text{ else } (\text{fun } y \rightarrow y)) 7) : \text{int}
 \end{array}$$

CS 421 Midterm 1

**Name:** \_\_\_\_\_

Worksheet (If extra space is needed).



CS 421 Midterm 1

**Name:** \_\_\_\_\_

Worksheet (If extra space is needed).





CS 421 Midterm 1

Name: \_\_\_\_\_

**Rules for type derivations:**

Constants:

$$\frac{}{\Gamma \vdash n : \text{int}} \quad (\text{assuming } n \text{ is an integer constant})$$

$$\frac{}{\Gamma \vdash \text{true} : \text{bool}} \quad \frac{}{\Gamma \vdash \text{false} : \text{bool}}$$

Variables:

$$\frac{}{\Gamma \vdash x : \sigma} \quad \text{if } \Gamma(x) = \sigma$$
Primitive operators ( $\oplus \in \{+, -, *, \text{mod}, \dots\}$ ):
$$\frac{\Gamma \vdash e_1 : \text{int} \quad \Gamma \vdash e_2 : \text{int}}{\Gamma \vdash e_1 \oplus e_2 : \text{int}}$$
Relations ( $\sim \in \{<, >, =, <=, >= \}$ ):
$$\frac{\Gamma \vdash e_1 : \text{int} \quad \Gamma \vdash e_2 : \text{int}}{\Gamma \vdash e_1 \sim e_2 : \text{bool}}$$

Connectives :

$$\frac{\Gamma \vdash e_1 : \text{bool} \quad \Gamma \vdash e_2 : \text{bool}}{\Gamma \vdash e_1 \ \&\& \ e_2 : \text{bool}} \quad \frac{\Gamma \vdash e_1 : \text{bool} \quad \Gamma \vdash e_2 : \text{bool}}{\Gamma \vdash e_1 \ \|\ \ e_2 : \text{bool}}$$

If\_then\_else rule:

$$\frac{\Gamma \vdash e_1 : \text{bool} \quad \Gamma \vdash e_2 : \tau \quad \Gamma \vdash e_3 : \tau}{\Gamma \vdash (\text{if } e_1 \text{ then } e_2 \text{ else } e_3) : \tau}$$

Application rule:

$$\frac{\Gamma \vdash e_1 : \tau_1 \rightarrow \tau_2 \quad \Gamma \vdash e_2 : \tau_1}{\Gamma \vdash (e_1 \ e_2) : \tau_2}$$

Function rule:

$$\frac{[x : \tau_1] + \Gamma \vdash e : \tau_2}{\Gamma \vdash \text{fun } x \rightarrow e : \tau_1 \rightarrow \tau_2}$$

Let rule:

$$\frac{\Gamma \vdash e_1 : \tau_1 \quad [x : \tau_1] + \Gamma \vdash e_2 : \tau_2}{\Gamma \vdash (\text{let } x = e_1 \text{ in } e_2) : \tau_2}$$

Let Rec rule:

$$\frac{[x : \tau_1] + \Gamma \vdash e_1 : \tau_1 \quad [x : \tau_1] + \Gamma \vdash e_2 : \tau_2}{\Gamma \vdash (\text{let rec } x = e_1 \text{ in } e_2) : \tau_2}$$